

Anharmonic oscillations in a molecular system.

Start with a potential that has a repulsive core and a longer range attraction.


Note that I have chosen constant values to make the arithmetic easy and to produce an angular frequency of unity.

```
> V:=1/(2*x^2)-1/x;
```

$$V := \frac{1}{2x^2} - \frac{1}{x}$$

Here is a view of the potential showing clearly the minimum centered at x=1.

```
> plot(V, x=0.4..4);
```



The force is negative the derivative of the potential

```
> F:=-diff(V, x);
```

$$F := \frac{1}{x^3} - \frac{1}{x^2}$$

so that the differential equation for a particle of mass 1 moving in the potential is

```
> eqn:=diff(x(t),t,t)=subs(x=x(t),F);
```

$$eqn := \frac{d^2}{dt^2} x(t) = \frac{1}{x(t)^3} - \frac{1}{x(t)^2}$$

Now use Maple's numeric differential equation solver to give us an accurate solution for the motion assuming that we start from rest near equilibrium. I have selected a starting point at $x=0.9$, 1/10 of the way to the origin from equilibrium. I told it to give us a solution valid for the first 30 seconds.

```
> s1:=dsolve({eqn,x(0)=0.9,D(x)(0)=0},x(t),range=0..30,numeric);  
s1 := proc(x_rkf45 ) ... end proc;
```


The object returned is a function that Maple can use to generate numbers. Such solutions can be graphed using a routine in the "plots" package, which we now load.

```
> with(plots):
```

```
Warning, the name changecoords has been redefined
```

odeplot will plot the solution using a few selected points. We want to get a nice smooth curve so I use the numpoints option to make sure that we get enough points for a pretty graph.

```
> odeplot(s1,numpoints=1000);
```




So we got four complete periods in about 25.7s corresponding to a frequency of 0.1556Hz or an angular frequency of 0.978s.

Since we were expecting an angular frequency of 1 this is a little bit low. Let's reduce the amplitude still further.


```
> s2:=dsolve({eqn,x(0)=0.99,D(x)(0)=0},x(t),range=0..30,numeric);
           s2 := proc(x_rkf45 ) ... end proc;
```

```
> odeplot(s2,numpoints=1000);
```




Now we have pulled the time for periods down to a hair over 25s compared to the theoretical value $4T=8\pi/\Omega=25.13$ s. So with amplitude of 1% of the equilibrium separation we are extremely close to harmonic. By 10% amplitude we saw that the frequency was about 2% low. Increasing to 50% amplitude we have

```
> s3:=dsolve({eqn,x(0)=0.5,D(x)(0)=0},x(t),range=0..30,numeric);
           s3 := proc(x_rkf45 ) ... end proc;
> odeplot(s3,numpoints=1000);
```



Well, that has made a rather large difference! Looking back at the original potential we see that a starting value of $x=0.5$ corresponds to an unbound solution and so perhaps this result is not so surprising. Let's look somewhere in between. At a starting amplitude of 25% of the equilibrium separation

```
> s4:=dsolve({eqn, x(0)=0.75, D(x)(0)=0}, x(t), range=0..30, numeric);
           s4 := proc(x_rkf45 ) ... end proc;
> odeplot(s4, numpoints=1000);
```



Two things are clear at this point. First the period has increased to 30s for four periods, an increase of 20% compared to the theoretical value. Second the shape is becoming pronouncedly non-sinusoidal. The tops of the peaks, the times when the particle is far out in the well, are much more rounded than the bottoms. The particle spends a much larger fraction of its time out beyond equilibrium than it does close in. This is precisely what we predicted in class on the basis of the shape of the potential curve.